

```
10 REM ***** DER HUNGRIGE FUCHS *****
20 REM Copyright 1985 by Berthold Freier und IDEA Verlag GmbH
30 REM
40 GOSUB 10000 : DEFINT A-Q,S-Z : GOSUB 1900
100 REM ***** Zueingabe *****
110 Z=0
120 INK 3,26,6
130 C$=INKEY$:IF C$="" THEN GOTO 130
140 C=VAL(C$) :Z=Z+1
150 A(HI,HJ)=0
160 IF C=1 THEN HI=HI-1:HJ=HJ+1:GOTO 240
170 IF C=3 THEN HI=HI+1:HJ=HJ+1:GOTO 240
180 IF C=9 THEN HI=HI+1:HJ=HJ-1:GOTO 240
190 IF C=7 THEN HI=HI-1:HJ=HJ-1:GOTO 240
200 IF C=2 THEN HJ=HJ+1: GOTO 240
210 IF C=8 THEN HJ=HJ-1: GOTO 240
220 IF C=4 THEN HI=HI-1: GOTO 240
230 IF C=6 THEN HI=HI+1: GOTO 240
240 I=HI : J=HJ : INK 3,6
300 REM ***** Zugpruefung *****
310 IF A(I,J)=0 THEN A(I,J)=5:GOTO 400
320 IF A(I,J)=1 THEN m$="Du bist im Stall gefangen !":GOSUB 1800:GOTO 700
330 IF A(I,J)=2 THEN m$="Du bist in eine Falle geraten !" : GOSUB 1800 :GOTO 700
```

```

340 IF A(I,J)=3 THEN A(I,J)=5 : KD=KD+1 : NS=NS-1 : GOSUB 1500 : GOTO 360
350 IF A(I,J)=4 THEN GOTO 370
360 IF NS=0 THEN M$="Du hast die letzte Gans gefressen ":GOSUB 1600: GOTO 700:EL
SE GOTO 400
370 IF RND<0.5 THEN GOTO 390
380 M$="Der Hund hat Dich erwischt !" : GOSUB 1800 :GOTO 700
390 A(I,J)=5 : ND=ND-1
400 IF Z<2 THEN GOSUB 1300 : GOTO 120 ELSE GOSUB 1300
500 REM ***** CPC 464 Bewegungen *****
510 Q=Q+1
520 FOR I=32 TO Q STEP -1: FOR J=3 TO 23
530 IF A(I,J)=0 OR A(I,J)=1 OR A(I,J)=2 OR A(I,J)=5 THEN GOTO 670
540 IF A(I,J)=3 THEN GOTO 610
550 K=I+(INT(RND*3-1))
560 IF K>31 OR K<2 THEN GOTO 550
570 L=J+(INT(RND*3-1))
580 IF L>23 OR L<3 THEN GOTO 570
590 IF A(K,L)=5 THEN M$="Der Hund hat Dich erwischt !" :GOSUB 1800: GOTO 700

600 IF A(K,L)<>0 THEN GOTO 550 ELSE A(K,L)=4 : A(I,J)=0 : GOTO 670
610 IF A(I+1,J)=5 THEN A(I,J)=0 : KD=KD+1 : GOSUB 1500:NS=NS-1 :GOTO 620 ELSE GO
TO 630
620 IF NS=0 THEN M$="Du hast die letzte Gans gefressen!": GOSUB 1600:GOTO 700 EL
SE GOTO 670
630 IF A(I+1,J)=2 THEN A(I,J)=0 : A(I+2,J)=3 : GOTO 670
640 IF I=31 AND A(I,J)=3 THEN A(I,J)=0 : ES=ES+1 : NS=NS-1 : GOTO 650 ELSE GOTO
660
650 IF NS=0 THEN M$="Die letzte Gans ist entkommen !":GOTO 700
660 A(I+1,J)=A(I,J):A(I,J)=0
670 NEXT J : NEXT I
680 INK 3,6 :GOSUB 1100 : GOTO 100
700 REM ***** Schlussanzeige *****
710 CLS
720 LOCATE 3,5:PEN 2:PRINT "Das Spiel ist beendet !"
730 LOCATE 3,8:PRINT M$
740 LOCATE 1,16 : PRINT "Willst Du es noch einmal versuchen ?"
750 LOCATE 12,18 : PRINT "( J / N )"
760 E$=INKEY$ : IF E$="" THEN GOTO 760
770 E$=UPPER$(E$)
780 IF E$="J" THEN RUN
790 IF E$="N" THEN LOCATE 7,20 : PEN 3
800 PRINT "Auf Wiedersehen !"
810 FOR n=1 TO 1000 : NEXT n :CLS : PEN 1: END
900 REM ***** Spielfeld *****
910 CLS
920 INK 2,17
930 PEN 2
940 FOR n=2 TO 24
950 LOCATE 1,n : PRINT CHR$(203)
960 LOCATE 32,n : PRINT STRING$(8,CHR$(206))
970 NEXT n
980 LOCATE 1,2 : PRINT STRING$(31,CHR$(203))
990 LOCATE 1,24 : PRINT STRING$(31,CHR$(203))
1000 PEN 1
1010 LOCATE 1,25 : PRINT "Fuchs:":PEN 3:PRINT CHR$(232)
1020 PEN 1
1030 LOCATE 15,25 : PRINT "Gans:":CHR$(230)
1040 LOCATE 30,25 : PRINT "Hund:":CHR$(231)
1050 LOCATE 1,1 : PRINT "Gänse gefressen:          entkommen:"
1060 RETURN
1100 REM ***** Spielfiguren *****
1110 FOR I=2 TO 31 : FOR J=3 TO 23
1120 IF A(I,J)=0 THEN w=128 : z=1 : GOTO 1170
1130 IF A(I,J)=2 THEN w=203 : z=2 : GOTO 1170
1140 IF A(I,J)=3 THEN w=230 : z=1 : GOTO 1170
1150 IF A(I,J)=4 THEN w=231 : z=1 : GOTO 1170
1160 IF A(I,J)=5 THEN w=232 : z=3 : GOTO 1170
1170 LOCATE I,J : PEN z : PRINT CHR$(w)

```

```

1180 NEXT J : NEXT I
1190 LOCATE 18,1:PRINT KD
1200 LOCATE 37,1:PRINT ES
1210 RETURN
1300 REM ***** Spielfiguren um Fuchs **
1310 FOR I=HI-1 TO HI+1 : FOR J=HJ-1 TO HJ+1
1320 IF A(I,J)=0 THEN w=128 : z=1 : GOTO 1390
1330 IF A(I,J)=1 AND I>=31 THEN w=206 : z=2 : GOTO 1390
1340 IF A(I,J)=1 AND I<31 THEN w=203 : z=2 : GOTO 1390
1350 IF A(I,J)=2 THEN w=203 : z=2 : GOTO 1390
1360 IF A(I,J)=3 THEN w=230 : z=1 : GOTO 1390
1370 IF A(I,J)=4 THEN w=231 : z=1 : GOTO 1390
1380 IF A(I,J)=5 THEN w=232 : z=3 : GOTO 1390
1390 LOCATE I,J: PEN z : PRINT CHR$(w)
1400 NEXT J : NEXT I
1410 LOCATE 18,1:PRINT KD
1420 LOCATE 37,1:PRINT ES
1430 RETURN
1500 REM ***** Fuchs erfolgreich *****
1510 SOUND 1,319,25
1520 SOUND 1,213,25
1530 SOUND 1,284,25
1540 SOUND 1,213,25
1550 SOUND 1,246,50
1560 SOUND 1,213,50
1570 RETURN
1600 REM ***** Siegmelodie *****
1610 SOUND 1,319,25
1620 SOUND 1,246,25
1630 SOUND 1,213,25
1640 SOUND 1,239,25
1650 SOUND 1,246,25
1660 SOUND 1,319,25
1670 SOUND 1,284,25
1680 SOUND 1,246,25
1690 SOUND 1,213,25
1700 SOUND 1,239,50
1710 SOUND 1,284,100
1720 RETURN
1800 REM ***** Verloren *****
1810 SOUND 1,478,25
1820 SOUND 1,284,25
1830 SOUND 1,478,25
1840 SOUND 1,284,25
1850 SOUND 1,426,25
1860 SOUND 1,253,25
1870 SOUND 1,478,50
1880 SOUND 1,284,50
1890 RETURN
1900 REM ***** Spielanleitung *****
1910 CLS : MODE 1 : INK 3,6
1920 LOCATE 5,1 :PEN 3: PRINT "D E R H U N G R I G E F U C H S":PEN 1
1930 LOCATE 1,3
1940 PRINT "Bei diesem Spiel bist Du der listige Fuchs, der sich sein Fressen
suchen muß und zu diesem Zweck in ein Gehege ein- dringt."
1950 PRINT : PRINT "Im Spiel sind Gänse, die Du fressen kannst, aber auch Hun-
de, die Deine An- griffe zu 50 % abwehren und Dich dabei töten."
1960 PRINT : PRINT "Außerdem gibt es den Stall und Fallen, aus denen es für Dic-
h kein Entrinnen gibt. Paße also gut auf!"
1970 PRINT : PRINT "Da Du schnell rennen kannst, darfst Du 2 Züge eingeben, bevo-
r der CPC 464 die Gänse und Hunde bewegt."
1980 PRINT : PRINT "Die Steuerung erfolgt über den Zehner- block mit 1 bis 4 un-
d 6 bis 9."
1990 PRINT:PRINT "Ich wünsche Dir viel Glück !"
2000 PEN 2:LOCATE 1,25:PRINT "Bitte eine Taste drücken !":PEN 1
2010 f$=INKEY$
2020 IF f$="" THEN GOTO 2010
2030 DIM A(33,26)
2040 RANDOMIZE TIME

```

```

2100 REM ***** Spielbedingungen *****
2110 CLS
2120 LOCATE 3,5 : INPUT "Wieviele Fallen ";NT
2130 IF NT>25 THEN LOCATE 1,7 : PRINT "Soll die Futtersuche so gefährlich sein?"
:GOTO 2150
2140 IF NT<1 THEN LOCATE 1,7 : PRINT "So wenig ?":GOTO 2150 ELSE GOTO 2160
2150 FOR l=1 TO 1000 : NEXT l : GOTO 2110
2160 LOCATE 3,9 : PRINT "Also, ";NT;" Fallen."
2170 FOR l=1 TO 1000 : NEXT l
2180 CLS
2190 LOCATE 3,5 : INPUT "Wieviele Gänse";NS
2200 IF NS<100 THEN LOCATE 1,7 : PRINT "Das verbieten die Schutzvorschriften !":
GOTO 2220
2210 IF NS<1 THEN LOCATE 1,7 : PRINT "Machst Du eine Hungerkur ?":GOTO 2220 ELSE
GOTO 2230
2220 FOR l=1 TO 1000 : NEXT l : GOTO 2180
2230 LOCATE 3,9 : PRINT "Also, ";NS;" Gänse."
2240 FOR l=1 TO 1000 : NEXT l
2250 CLS
2260 LOCATE 3,5 : INPUT "Wieviele Hunde";ND
2270 IF ND>25 THEN LOCATE 1,7 : PRINT "So viele Hunde hat kein Bauer !":GOTO 22
90
2280 IF ND<1 THEN LOCATE 1,7 : PRINT "Füchse sind nicht vom Aussterben bedroht":
GOTO 2290 ELSE GOTO 2300
2290 FOR l=1 TO 1000 : NEXT l : GOTO 2250
2300 LOCATE 3,9 : PRINT "Also, ";ND;" Hunde."
2310 FOR l=1 TO 1000 : NEXT l
2320 GOSUB 900
2400 REM ***** Initialisierung *****
2410 REM ***** Fallen *****
2420 FOR f=1 TO NT
2430 I=INT(RND*30)+2
2440 J=INT(RND*21)+3
2450 IF A(I,J)<>0 THEN GOTO 2430 ELSE A(I,J)=2
2460 NEXT f
2470 REM ***** Gänse *****
2480 FOR s=1 TO NS
2490 I=INT(RND*28)+3
2500 J=INT(RND*21)+3
2510 IF A(I,J)<>0 THEN GOTO 2490 ELSE A(I,J)=3
2520 NEXT s
2530 REM ***** Rand *****
2540 FOR n=2 TO 26
2550 A(1,n)=2 : A(32,n)=1 : A(33,n)=1
2560 NEXT n
2570 REM ***** Hunde *****
2580 FOR t=1 TO ND
2590 I=INT(RND*30)+3
2600 J=INT(RND*21)+3
2610 IF A(I,J)<>0 THEN GOTO 2590 ELSE A(I,J)=4
2620 NEXT t
2630 REM ***** Rand *****
2640 FOR p=1 TO 32
2650 A(p,2)=2 : A(p,24)=2
2660 NEXT p
2670 REM ***** Fuchs *****
2680 I=INT(RND*7)+24
2690 J=INT(RND*21)+3
2700 IF A(I,J)<>0 THEN GOTO 2590 ELSE A(I,J)=5
2710 HI=I
2720 HJ=J
2730 Q=1
2740 GOSUB 1100
2750 RETURN

```